
ΑΡΧΗ 1ΗΣ ΣΕΛΙΔΑΣ  

ΗΜΕΡΗΣΙΩΝ ΚΑΙ ΕΣΠΕΡΙΝΩΝ ΓΕΝΙΚΩΝ ΛΥΚΕΙΩΝ 

ΤΕΛΟΣ 1ΗΣ ΑΠΟ 4  ΣΕΛΙΔΕΣ 

ΠΑΝΕΛΛΑΔΙΚΕΣ ΕΞΕΤΑΣΕΙΣ 
ΗΜΕΡΗΣΙΩΝ & ΕΣΠΕΡΙΝΩΝ ΓΕΝΙΚΩΝ ΛΥΚΕΙΩΝ 

ΠΕΜΠΤΗ 8 ΙΟΥΝΙΟΥ 2023 
ΕΞΕΤΑΖΟΜΕΝΟ ΜΑΘΗΜΑ:  ΠΛΗΡΟΦΟΡΙΚΗ ΠΡΟΣΑΝΑΤΟΛΙΣΜΟΥ  

ΣΥΝΟΛΟ ΣΕΛΙΔΩΝ: ΤΕΣΣΕΡΙΣ (4) 
 

ΘΕΜΑ Α 

Α1. Να γράψετε στο τετράδιό σας τον αριθμό καθεμιάς από τις παρακάτω 
προτάσεις 1 έως 5 και δίπλα τη λέξη ΣΩΣΤΟ, αν η πρόταση είναι σωστή ή 
τη λέξη ΛΑΘΟΣ, αν η πρόταση είναι λανθασμένη. 

1. Οποιαδήποτε εντολή επανάληψης ΟΣΟ…ΕΠΑΝΑΛΑΒΕ μπορεί να 
μετατραπεί σε εντολή επανάληψης ΓΙΑ…ΑΠΟ…ΜΕΧΡΙ…ΜΕ_ΒΗΜΑ. 

2. Η μέθοδος επεξεργασίας «πρώτο μέσα πρώτο έξω» (FIFO) 
εφαρμόζεται στη δομή δεδομένων ΟΥΡΑ. 

3. Κάθε συνάρτηση επιστρέφει μόνο μια τιμή. 
4. Η έκφραση A MOD 5 είναι συντακτικά σωστή στη ΓΛΩΣΣΑ, όταν το Α 

είναι  πραγματική μεταβλητή. 
5. Σε μια λίστα τα στοιχεία δεν μπορούν να προστεθούν στην αρχή ή στο 

τέλος της, παρά μόνο στη μέση της. 
Μονάδες 10 

Α2.   Δίνεται το παρακάτω δυαδικό δένδρο, στο οποίο εμφανίζονται πέντε κενοί 
κόμβοι: Κ1, Κ2, Κ3, Κ4, Κ5.  

 Για καθέναν από τους κόμβους να γράψετε στο τετράδιό σας τα Κ1, Κ2, Κ3, 
Κ4, Κ5 και δίπλα την κατάλληλη τιμή από τις τιμές: 4, 6, 15, 20, 34, ώστε 
το δένδρο να είναι δυαδικό δένδρο αναζήτησης. 

 

 
Μονάδες 5 

Α3. Να αναφέρετε τις βασικές πράξεις των συνδεδεμένων λιστών. 

Μονάδες 5 

Α4. Να αναφέρετε επιγραμματικά τα κριτήρια που πρέπει να ικανοποιεί ένας 
αλγόριθμος.  

Μονάδες 5 
 



ΑΡΧΗ 2ΗΣ ΣΕΛΙΔΑΣ  

ΗΜΕΡΗΣΙΩΝ ΚΑΙ ΕΣΠΕΡΙΝΩΝ ΓΕΝΙΚΩΝ ΛΥΚΕΙΩΝ 

ΤΕΛΟΣ 2ΗΣ ΑΠΟ 4  ΣΕΛΙΔΕΣ 

 
ΘΕΜΑ Β  

Β1. Δίνεται το παρακάτω τμήμα αλγορίθμου:  
 

 ΓΙΑ  i  ΑΠΟ Α ΜΕΧΡΙ  Μ  ΜΕ_ΒΗΜΑ  Β 
   ΓΡΑΨΕ  i 
 ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 

 Για καθεμιά από τις παρακάτω τρεις περιπτώσεις τιμών των μεταβλητών 
Α, Μ, Β να γράψετε στο τετράδιό σας τον αριθμό της περίπτωσης και δίπλα 
πόσες φορές θα εκτελεστεί η εντολή ΓΡΑΨΕ. 

1. Α= 2 Μ= 0  Β= -1 
2. Α= 5 Μ= 0  Β= 2 
3. Α= -3 Μ= 3  Β= 2          Μονάδες 6 

Β2. Δίνονται τα παρακάτω ζεύγη: 

 

 Υπερκλάση Υποκλάση 

1 Τράπεζα Λογαριασμός 

2 Δήμος Συνοικία 

3 Μέσο μετακίνησης Ποδήλατο 

4 Γεωμετρικό σχήμα Τετράγωνο 

5 Σχολείο Σχολική Τάξη 

 

 Για κάθε ζεύγος να γράψετε στο τετράδιό σας τον αριθμό του (1 έως 5) 
και δίπλα τη λέξη ΝΑΙ, εάν πρόκειται για έγκυρο ζεύγος Υπερκλάσης – 
Υποκλάσης ή τη λέξη ΟΧΙ σε αντίθετη περίπτωση. 

Μονάδες 5 

Β3. Έστω στοίβα που υλοποιείται με μονοδιάστατο πίνακα Σ[Ν] και ουρά που 
υλοποιείται με μονοδιάστατο πίνακα Ο[Ν]. Για τη διαχείριση των 
λειτουργιών της στοίβας χρησιμοποιείται μεταβλητή με όνομα top, ενώ για 
τη διαχείριση των λειτουργιών της ουράς χρησιμοποιούνται μεταβλητές με 
ονόματα front και rear. Δίνονται στη συνέχεια κάποιες περιπτώσεις 
ελέγχων που αφορούν στη στοίβα και στην ουρά. Για καθέναν από τους 
ελέγχους να γράψετε στο τετράδιό σας τον αριθμό του (1 έως 4) και δίπλα 
τη συνθήκη που υλοποιεί τον αντίστοιχο έλεγχο σε ψευδογλώσσα. 

1. Η στοίβα είναι άδεια. 
2. Η ουρά είναι γεμάτη. 
3. Η στοίβα έχει ένα στοιχείο. 
4. Η ουρά έχει δύο στοιχεία. 

Μονάδες 4 
 

Β4. Το παρακάτω τμήμα αλγορίθμου διαβάζει επαναληπτικά αριθμούς και 
υπολογίζει το άθροισμα των θετικών, ενώ τερματίζει τις επαναλήψεις σε 
οποιαδήποτε από τις εξής περιπτώσεις: 
- όταν διαβαστούν 100 αριθμοί. 
- όταν διαβαστούν διαδοχικά τρεις αρνητικοί αριθμοί. 



ΑΡΧΗ 3ΗΣ ΣΕΛΙΔΑΣ  

ΗΜΕΡΗΣΙΩΝ ΚΑΙ ΕΣΠΕΡΙΝΩΝ ΓΕΝΙΚΩΝ ΛΥΚΕΙΩΝ 

ΤΕΛΟΣ 3ΗΣ ΑΠΟ 4  ΣΕΛΙΔΕΣ 

 
Σ  0 ! άθροισμα των θετικών 
π  0 ! πλήθος αριθμών που διαβάστηκαν 
π_α  0 ! πλήθος αρνητικών αριθμών που διαβάστηκαν διαδοχικά 
ΟΣΟ π_α < 3   …(1)…  π < 100  ΕΠΑΝΑΛΑΒΕ 
 π  …(2)…   
 ΔΙΑΒΑΣΕ x 
 ΑΝ x > 0 ΤΟΤΕ  
  Σ  Σ + x 
  π_α  …(3)…   
 ΑΛΛΙΩΣ_ΑΝ x < 0 τότε  
  π_α  …(4)…    
 ΑΛΛΙΩΣ 
  π_α  …(5)…   
 ΤΕΛΟΣ_ΑΝ 
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ   
 

 Για καθένα από τα κενά (1 έως 5) να γράψετε στο τετράδιό σας τον αριθμό του 
και δίπλα ό,τι χρειάζεται να συμπληρωθεί, ώστε να υλοποιείται σωστά η 
λειτουργία που περιγράφηκε.          Μονάδες 10 

 
ΘΕΜΑ Γ 

Μια εταιρεία κινητής τηλεφωνίας καταγράφει τη διάρκεια των τηλεφωνικών 
κλήσεων σε δευτερόλεπτα αλλά χρεώνει ολόκληρα λεπτά από το πρώτο 
δευτερόλεπτο. Όταν η διάρκεια ομιλίας είναι από 1 έως 60 δευτερόλεπτα χρεώνει 
ένα ολόκληρο λεπτό της ώρας, όταν είναι από 61 έως 120 δευτερόλεπτα χρεώνει 
δύο ολόκληρα λεπτά της ώρας και ούτω καθεξής. 

Σε κάθε κλήση τα τρία πρώτα λεπτά της ώρας χρεώνονται 0.06 ευρώ το λεπτό, 
ενώ η διάρκεια επιπλέον των τριών λεπτών χρεώνεται 0.04 ευρώ το λεπτό 
(κλιμακωτή χρέωση).  

Να κατασκευάσετε πρόγραμμα σε ΓΛΩΣΣΑ, το οποίο: 

Γ1. Να περιλαμβάνει κατάλληλο τμήμα δηλώσεων.       Μονάδες 2 

Γ2. Για κάθε κλήση να ζητάει τη διάρκεια ομιλίας σε δευτερόλεπτα ελέγχοντας 
ότι δίνεται θετικός αριθμός και να εμφανίζει τη χρέωσή της. Ο 
υπολογισμός της χρέωσης να γίνεται με κλήση της συνάρτησης ΧΡΕΩΣΗ 
που περιγράφεται στο ερώτημα Γ5.        Μονάδες 5 

Γ3. Η παραπάνω διαδικασία να τερματίζεται σε οποιαδήποτε από τις εξής 
περιπτώσεις: 
 όταν το σύνολο των χρεώσεων ξεπεράσει τα 10 ευρώ. 
 όταν συμπληρωθούν 100 κλήσεις. 

Μονάδες 6 
Γ4. Μετά το τέλος της επαναληπτικής διαδικασίας να εμφανίζει το επί τοις 

εκατό ποσοστό των κλήσεων με χρέωση από 2 ευρώ και πάνω. 
Μονάδες 6 

Γ5. Να κατασκευάσετε τη συνάρτηση ΧΡΕΩΣΗ, η οποία να δέχεται ως είσοδο 
τη διάρκεια κλήσης σε δευτερόλεπτα, να υπολογίζει τα λεπτά της ώρας 
που θα χρεώσει και να επιστρέφει την αντίστοιχη χρέωση.      Μονάδες 6 

 
ΣΗΜΕΙΩΣΗ: Να θεωρήσετε ότι η διάρκεια ομιλίας εισάγεται ως ακέραια τιμή. 



ΑΡΧΗ 4ΗΣ ΣΕΛΙΔΑΣ  

ΗΜΕΡΗΣΙΩΝ ΚΑΙ ΕΣΠΕΡΙΝΩΝ ΓΕΝΙΚΩΝ ΛΥΚΕΙΩΝ 

ΤΕΛΟΣ 4ΗΣ ΑΠΟ 4  ΣΕΛΙΔΕΣ 

 
ΘΕΜΑ Δ 
 
Ένας τουριστικός όμιλος διαθέτει 10 ξενοδοχεία. 

Να κατασκευάσετε πρόγραμμα σε ΓΛΩΣΣΑ, το οποίο:  

Δ1. α) Να περιλαμβάνει κατάλληλο τμήμα δηλώσεων (μονάδες 2). 
  β) Για κάθε ξενοδοχείο να διαβάζει το όνομά του σε πίνακα ON[10] καθώς 

και τον μηνιαίο αριθμό επισκεπτών για κάθε μήνα του έτους σε πίνακα 
ΕΠ[10,12] (μονάδες 3). 

Μονάδες 5 

Δ2. Για κάθε μήνα να εμφανίζει τον αριθμό του μήνα (1 έως 12) και δίπλα: 
 το πλήθος των ξενοδοχείων που είχαν περισσότερους από 1000 

επισκέπτες  ή 
 το μήνυμα: «ΚΑΝΕΝΑ ΞΕΝΟΔΟΧΕΙΟ». 

Μονάδες 5 

Δ3. Να εμφανίζει τα ονόματα των ξενοδοχείων με τον μικρότερο συνολικό 
αριθμό επισκεπτών στη διάρκεια του έτους.       Μονάδες 9 

Δ4. Να εμφανίζει τα ονόματα των ξενοδοχείων και τον ετήσιο αριθμό 
επισκεπτών κατά φθίνουσα σειρά ετήσιου αριθμού επισκεπτών. Σε 
περίπτωση που δύο ή περισσότερα ξενοδοχεία έχουν τον ίδιο ετήσιο 
αριθμό επισκεπτών να εμφανίζει τα ονόματα των ξενοδοχείων 
αλφαβητικά. 

Μονάδες 6 
 

ΣΗΜΕΙΩΣΗ: Να θεωρήσετε ότι όλα τα ξενοδοχεία είχαν επισκέπτες στη διάρκεια 
του χρόνου. 

 
 

ΟΔΗΓΙΕΣ (για τους  εξεταζομένους /τις  εξεταζόμενες) 
 

1. Στο εξώφυλλο του τετραδίου να γράψετε το εξεταζόμενο μάθημα. Στο  
εσώφυλλο  πάνω-πάνω να συμπληρώσετε τα ατομικά στοιχεία μαθητή. 
Στην  αρχή των απαντήσεών σας  να γράψετε πάνω-πάνω την 
ημερομηνία  και το  εξεταζόμενο μάθημα . Να  μην  αντιγράψετε τα θέματα  
στο  τετράδιο και να  μη  γράψετε  πουθενά στις απαντήσεις σας το  όνομά  
σας. 

2. Να γράψετε το ονοματεπώνυμό σας στο πάνω μέρος των φωτοαντιγράφων 
αμέσως μόλις σας παραδοθούν. Τυχόν σημειώσεις σας πάνω στα θέματα 
δεν θα βαθμολογηθούν σε καμία περίπτωση. Κατά  την αποχώρησή  σας 
να παραδώσετε μαζί με το τετράδιο και  τα φωτοαντίγραφα. 

3. Να απαντήσετε στο  τετράδιό  σας σε όλα τα θέματα μόνο  με μπλε ή μόνο  
με  μαύρο στυλό με μελάνι  που δεν σβήνει. Μολύβι επιτρέπεται, μόνο αν 
το  ζητάει  η  εκφώνηση, και  μόνο  για πίνακες, διαγράμματα  κλπ . 

4. Κάθε απάντηση επιστημονικά  τεκμηριωμένη είναι  αποδεκτή. 
5. Διάρκεια εξέτασης: τρεις (3) ώρες μετά τη διανομή  των φωτοαντιγράφων.  
6. Ώρα δυνατής αποχώρησης:  10.00 π.μ. 

ΣΑΣ  ΕΥΧΟΜΑΣΤΕ KΑΛΗ ΕΠΙΤΥΧΙΑ  
ΤΕΛΟΣ ΜΗΝΥΜΑΤΟΣ 



ΑΝΑΠΤΥΞΗ ΕΦΑΡΜΟΓΩΝ ΣΕ ΠΡΟΓΡΑΜΜΑΤΙΣΤΙΚΟ ΠΕΡΙΒΑΛΛΟΝ 

ΤΕΛΟΣ 1ης ΑΠΟ 4 ΣΕΛΙΔΕΣ 

ΕΝΔΕΙΚΤΙΚΕΣ ΑΠΑΝΤΗΣΕΙΣ 

ΘΕΜΑ A  

Α1. 1. → Λ 

 2. → Σ 

 3. → Σ 

 4. → Λ 

 5. → Λ 

 

 

Α2. Κ1 → 20 

 Κ2 → 6 

 Κ3 → 4 

Κ4 → 15 

Κ5 → 34 

 

Α3. Εισαγωγή κόμβου στη λίστα. 

 Διαγραφή κόμβου από τη λίστα. 

  Έλεγχος για το αν η λίστα είναι κενή. 

  Αναζήτηση κόμβου για την εύρεση συγκεκριμένου στοιχείου. 

  Διάσχιση της λίστας και προσπέλαση των στοιχείων της. 

  

Α4. Είσοδος. Καμία, μία ή περισσότερες τιμές δεδομένων πρέπει να δίνονται ως είσοδοι στον αλγόριθμο.  

 Έξοδος. Ο αλγόριθμος πρέπει να δημιουργεί τουλάχιστον μία τιμή δεδομένων ως αποτέλεσμα προς 

το χρήστη ή προς έναν άλλο αλγόριθμο. 

 Καθοριστικότητα. Κάθε εντολή πρέπει να καθορίζεται χωρίς καμία αμφιβολία για τον τρόπο 

εκτέλεσής της.  

 Περατότητα. Ο αλγόριθμος να τελειώνει μετά από πεπερασμένα βήματα εκτέλεσης των εντολών του.  

 Αποτελεσματικότητα. Κάθε μεμονωμένη εντολή του αλγορίθμου να είναι απλή ώστε να μπορεί να 

εκτελεστεί.  

 

  

 

  



ΑΝΑΠΤΥΞΗ ΕΦΑΡΜΟΓΩΝ ΣΕ ΠΡΟΓΡΑΜΜΑΤΙΣΤΙΚΟ ΠΕΡΙΒΑΛΛΟΝ 

ΤΕΛΟΣ 2ης ΑΠΟ 4 ΣΕΛΙΔΕΣ 

ΘΕΜΑ B 

B1. 1. τρεις (3) 

 2. καμία (0) 

 3. τέσσερις (4) 

 

Β2. 1.  ΟΧΙ 

 2. ΟΧΙ 

 3. ΝΑΙ 

 4. ΝΑΙ 

 5. ΟΧΙ 

 

Β3. 1. top = 0   

 2. rear = N     

 3. top = 1 

 4. rear – front + 1 = 2   (ή  rear – front = 1  ή  rear = front + 1)  

 

B4. 1. KAI 

 2. π + 1 

 3. 0 

 4. π_α  + 1 

 5. 0 

  

  



ΑΝΑΠΤΥΞΗ ΕΦΑΡΜΟΓΩΝ ΣΕ ΠΡΟΓΡΑΜΜΑΤΙΣΤΙΚΟ ΠΕΡΙΒΑΛΛΟΝ 

ΤΕΛΟΣ 3ης ΑΠΟ 4 ΣΕΛΙΔΕΣ 

ΘΕΜΑ Γ 
ΠΡΟΓΡΑΜΜΑ ΘΕΜΑ_Γ 
ΜΕΤΑΒΛΗΤΕΣ 
  ΑΚΕΡΑΙΕΣ: C, C2, X 
  ΠΡΑΓΜΑΤΙΚΕΣ: S, ΧΡ 
ΑΡΧΗ 
  C <- 0 
  C2 <- 0 
  S <- 0 
  ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ !ή ΟΣΟ C < 100 ΚΑΙ S <= 10 ΕΠΑΝΑΛΑΒΕ 
  C <- C + 1 
     ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ 
      ΔΙΑΒΑΣΕ X 
     ΜΕΧΡΙΣ_ΟΤΟΥ X > 0 
     ΧΡ <- ΧΡΕΩΣΗ(X) 
     ΓΡΑΨΕ ΧΡ 
     S <- S + ΧΡ 
     ΑΝ ΧΡ >= 2 ΤΟΤΕ 
       C2 <- C2 + 1 
     ΤΕΛΟΣ_ΑΝ 
   ΜΕΧΡΙΣ_ΟΤΟΥ C = 100 Η S > 10 
   ΓΡΑΨΕ C2 / C * 100 
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ 
!============================== 
ΣΥΝΑΡΤΗΣΗ ΧΡΕΩΣΗ(sec):ΠΡΑΓΜΑΤΙΚΗ 
ΜΕΤΑΒΛΗΤΕΣ 
  ΑΚΕΡΑΙΕΣ: sec, min 
   ΠΡΑΓΜΑΤΙΚΕΣ: ΧΡ 
ΑΡΧΗ 
   min <- sec div 60 
  ΑΝ sec mod 60 <> 0 ΤΟΤΕ 
    min <- min + 1 
  ΤΕΛΟΣ_ΑΝ 
  ΑΝ min <= 3 ΤΟΤΕ 
    ΧΡ <- min * 0.06 
  ΑΛΛΙΩΣ 
    ΧΡ <- 3 * 0.06 + (min - 3) * 0.04 
  ΤΕΛΟΣ_ΑΝ 
  ΧΡΕΩΣΗ <- ΧΡ 
ΤΕΛΟΣ_ΣΥΝΑΡΤΗΣΗΣ 
  



ΑΝΑΠΤΥΞΗ ΕΦΑΡΜΟΓΩΝ ΣΕ ΠΡΟΓΡΑΜΜΑΤΙΣΤΙΚΟ ΠΕΡΙΒΑΛΛΟΝ 

ΤΕΛΟΣ 4ης ΑΠΟ 4 ΣΕΛΙΔΕΣ 

ΘΕΜΑ Δ 
ΠΡΟΓΡΑΜΜΑ ΘΕΜΑ_Δ 
!Δ1α 
ΜΕΤΑΒΛΗΤΕΣ 
  ΑΚΕΡΑΙΕΣ: i, j, ΕΠ[10, 12], C, S, SUM[10], min, temp1 
  ΧΑΡΑΚΤΗΡΕΣ: ON[10], temp2 
ΑΡΧΗ 
!Δ1β 
  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 
    ΔΙΑΒΑΣΕ ON[i] 
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 12 
      ΔΙΑΒΑΣΕ ΕΠ[i, j] 
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
!Δ2 
  ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 12 
    C <- 0 
    ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 
      ΑΝ C > 1000 ΤΟΤΕ 
        C <- C + 1 
      ΤΕΛΟΣ_ΑΝ 
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
    ΑΝ C > 0 ΤΟΤΕ 
      ΓΡΑΨΕ j, ':', C 
    ΑΛΛΙΩΣ 
      ΓΡΑΨΕ 'ΚΑΝΕΝΑ ΞΕΝΟΔΟΧΕΙΟ' 
    ΤΕΛΟΣ_ΑΝ 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
!Δ3 
   ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 
    S <- 0 
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 12 
      S <- S + ΕΠ[i, j] 
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
    SUM[i] <- S 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
  min <- SUM[1] 
  ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ 10 
    ΑΝ SUM[i] < min ΤΟΤΕ 
      min <- SUM[i] 
    ΤΕΛΟΣ_ΑΝ 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 
    ΑΝ SUM[i] = min ΤΟΤΕ 
      ΓΡΑΨΕ ON[i] 
    ΤΕΛΟΣ_ΑΝ 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
!Δ4 
  ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ 10 
    ΓΙΑ j ΑΠΟ 10 ΜΕΧΡΙ i ΜΕ_ΒΗΜΑ -1 
      ΑΝ SUM[j - 1] < SUM[j] ΤΟΤΕ 
        temp1 <- SUM[j - 1] 
        SUM[j - 1] <- SUM[j] 
        SUM[j] <- temp1 
        temp2 <- ON[j - 1] 
        ON[j - 1] <- ON[j] 
        ON[j] <- temp2 
      ΑΛΛΙΩΣ_ΑΝ SUM[j - 1] = SUM[j] ΤΟΤΕ 
        ΑΝ ON[j - 1] > ON[j] ΤΟΤΕ 
          temp2 <- ON[j - 1] 
          ON[j - 1] <- ON[j] 
          ON[j] <- temp2 
        ΤΕΛΟΣ_ΑΝ 
      ΤΕΛΟΣ_ΑΝ 
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
  ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10 
    ΓΡΑΨΕ ON[i], ':', SUM[i]  
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ     

 
Θα μπορούσε το Δ3 να γίνει μαζί με το Δ4 ως εξής: 
Αφού κάνουμε την ταξινόμηση ο μικρότερος συνολικός 
αριθμός επισκεπτών θα βρεθεί στην 10η θέση. 
Εμφανίζουμε λοιπόν το όνομα του 10ου ξενοδοχείου 
και πηγαίνοντας προς τα πάνω εμφανίζουμε και τα 
ονόματα όσων τυχόν έχουν τον ίδιο αριθμό 
επισκεπτών με το 10ο ξενοδοχείο. 
Κάτι τέτοιο θα μπορούσε να υλοποιηθεί με τις εντολές: 
 
!Β τρόπος για το Δ3 
  i <- 10 
  done <- ΑΛΗΘΗΣ !θα έπρεπε να δηλωθεί 
  ΟΣΟ done = ΑΛΗΘΗΣ ΚΑΙ i >= 1 ΕΠΑΝΑΛΑΒΕ 
    ΑΝ SUM[i] = SUM[10] ΤΟΤΕ 
      ΓΡΑΨΕ ON[i] 
    ΑΛΛΙΩΣ 
      done <- ΨΕΥΔΗΣ 
    ΤΕΛΟΣ_ΑΝ 
    i <- i - 1 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
 
  ! ή ακόμη πιο απλά 
  ΓΙΑ i ΑΠΟ 10 ΜΕΧΡΙ 1 ΜΕ_ΒΗΜΑ -1 
    ΑΝ SUM[i] = SUM[10] ΤΟΤΕ 
      ΓΡΑΨΕ ON[i] 
    ΤΕΛΟΣ_ΑΝ 
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ 
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ 

 

* 

* 

 




